# $Id$
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=xorg-xmag
pkgver=1.0.6
pkgrel=1
pkgdesc="Magnify parts of the screen"
arch=('i686' 'x86_64')
url="https://xorg.freedesktop.org/"
license=('custom')
depends=('libxaw' 'libxmu' 'libxt' 'libx11')
makedepends=('xorg-util-macros')
source=(https://xorg.freedesktop.org/archive/individual/app/xmag-${pkgver}.tar.bz2{,.sig})
sha256sums=('87a2bc23b251e2d8f8370d3283a4d6c8dac98a30cb5749a04336cdb55c14e161'
            'SKIP')
validpgpkeys=('4A193C06D35E7C670FA4EF0BA2FB9E081F2D130E') # Alan Coopersmith <alan.coopersmith@oracle.com>

build() {
  cd xmag-${pkgver}
  ./configure --prefix=/usr
  make
}

package() {
  cd xmag-${pkgver}
  make DESTDIR="${pkgdir}" install
  install -Dm644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/COPYING"
}
